
Detailed Methods 
Sample Preparation.  

A sample of L99A T4 Lysozyme was prepared according to previously published protocols1. 

The concentration of D2O was kept to a minimum (≈1%) to minimise the presence of 

isotopically shifted 15N resonances2. [13C6-15N4]-L-arginine hydrochloride was purchased from 

Sigma-Aldrich. 

 

NMR Spectroscopy.  

For all experiments, the spectrometer temperature was calibrated using a sample of d4-methanol 

and following standard protocols. 

Longitudinal EXSY3 zz-exchange experiments on free [13C6,15N4]-L-arginine were 

carried out and analysed as described previously1. All zz-exchange NMR spectra were recorded 

at a static magnetic field of 14.1 T on a Bruker Avance II spectrometer and using a room-

temperature TXI HCN inverse probe. In all of these experiments, 20 mixing times between 2 

ms and 300 ms were used, including two repeats for error estimation. The carrier frequencies 

were set to 4.7 ppm (1H) and 70.3 ppm (15N). The sweep width in the indirect dimension was 

set to 16.5 ppm with 40 complex points recorded. Four scans were recorded per transient and 

the recycle recovery delay (d1) was 1 s, giving a total recording time of ~2 hours per 

experiment. 

The presented MQ-CEST profiles were all recorded using the pulse sequence in Fig 1D 

(with 13C detection). The parameters employed in the MQ-CEST experiments are detailed in 

Table S1. The B1 field strengths and inhomogeneity were calibrated for each radio-frequency 

probe using the approach of Guenneugues et al4. 

 

Data analysis.  

Simulation and Fitting of MQ-CEST Data. For the simulation and fitting of MQ-CEST data 

in arginine guanidinium groups the three-spin system, C!" # N!$ %! N!$ %", is considered. The 

composite decoupling scheme ensures that the H! % nuclei have no significant effect on this 

spin system during the CEST period, although these protons are an important source of auto-

relaxation. All combinations of direct products of the individual spin product operators in 

C!" & N!$ %! N!$ %" and the identity operator give rise to 64 product operators5 required to fully 

describe this spin system. To account for the rotational exchange about the Cζ-Nε bond 

(symmetrical exchange of Nη1 and Nη2) a further 63 product operators are required: for 

convenience (in common with literature conventions) we can refer to the two states as the 

ground (G) and excited (E) state although the symmetric exchange here means that these states 

have equal energy and population. This gives a total of 127 product operators, describing the 

full spin system. In order to fit or simulate MQ-CEST data the Bloch-McConnell equation for 

the system6,7 needs to be solved, 

 
'
'(
𝐦(t) = −𝐋 ∙ 𝐦(t)         (1) 

 

Here, m(t) is an (n×1) column vector containing the n product operator terms for the spin 

system and L is the (n×n) Liouvillian matrix that describes its dynamics and time-dependence. 

To calculate the Liouvillian matrix we first write down the Hamiltonian for each spin 

interaction: in the present case the chemical shift of the two 15Nη nuclei, the CEST pulse and J-

couplings between the 15Nη nuclei and the 13Cζ nucleus are included. The small scalar-coupling 

between the two 15Nη is ignored8. For each Hamiltonian, 𝐻-, the commutator superoperator, 𝐻--, 

is calculated as: 

 

𝐻-- = 	𝐻- ⊗ 𝐸2 − 𝐸2 ⊗𝐻-3          (2) 

 

In which 𝐸2  is the identity matrix and the tilde denotes the complex conjugate. The elements of 

L can then be found by transforming 𝐻-- to the product operator basis9.  

Having added these interactions to the Liouvillian, relaxation terms are added to the 

matrix phenomenologically, following the approach of Helgstrand et al10. Due to the small 

gyromagnetic ratios of 13C and 15N dipolar interactions between them will be a relatively small 

source of relaxation11, which are ignored. Instead, the auto-relaxations will be dominated by 

dipolar relaxation to proximal 1H nuclei and the chemical shift anisotropy of the individual 

spins. Consequently, to a good approximation, we can write the spin relaxation of higher order 

product operators as the sum of the individual longitudinal and transverse single order spin 

operators: 

 

𝑅)#$*%&!*'&"
= 𝑅)#$

+ 𝑅*%
&! + 𝑅*'

&"            (3) 

Where 𝑗, 𝑘, 𝑙 ∈ (𝑥, 𝑦, 𝑧). It is also assumed that 𝑅*%
&! =	𝑅*%

&" . This approach considerably 

reduces the number of parameters to be optimised in the model and, as shown below, the 

derived kex and Δω values are relatively uncorrelated from the transverse relaxation rates (Fig 

S8). It should however be noted that the MQ-CEST approach should not be treated as a method 

of attaining relaxation rates of the various product operator terms. It should also be noted that 

the 15Nη nuclei will undergo multiexponential relaxation, which we approximate here with a 

single exponential, as a result of cross-correlated dipolar relaxation with each of the two 

directly bonded 1Hη nuclei12.  

To complete the Liouvillian matrix, L, chemical exchange between the two 15Nη nuclei 

needs to be added. The 127×127 exchange matrix k is defined as: 

 

𝐤 = >
0 0⋯0
0
⋮
0

@ 𝑘+, −𝑘,+
−𝑘+, 𝑘,+

A⊗ 𝐈-"
C,		            (4) 

 

where kGE and kEG are the forward and backwards rates of rotation, respectively, kex = kGE + kEG 

and 𝐈-" is the 63×63 identity matrix. Since the exchange between 15Nη1 and 15Nη2 is symmetric, 

kEG  = kGE. Finally, to obtain the full Liouvillian, L, k is added to Eq 2. 

To reduce the size of the Liouvillian matrix used to propagate the spin-system, it is 

noted that, in the MQ-CEST experiment, transverse 13Cζ magnetisation cannot be created and 

so any terms containing these are ignored. Removing these terms, we arrive at a 63×63 matrix, 

L, that describes the evolution of the three-spin 𝐶.𝑁/!𝑁/" spin system during the MQ-CEST 

experiment.  

Once L is known and since L is time-independent, Eq. 1 can be solved according to, 

 

𝐦(𝑡) = exp(−𝐋𝑡) ∙ 𝐦(0),         (5) 

 

where Eq. 5 in turn can be solved by directly taking the matrix exponential of L, using the Padé 

approximation (as implemented in the python module scipy13) or similar. For analysis of the 

MQ-CEST data, the matrix exponential can be calculated more rapidly through a Taylor 

expansion with scaling and squaring14. The initial condition of the column vector, m(0), is set 

to 4𝐶0
1𝑁2

/!𝑁2
/"(𝐺) = 𝑝+  and  4𝐶0

1𝑁2
/!𝑁2

/"(𝐸) = 	𝑝, with 𝑝+ = 𝑝, = 0.5 for the symmetrical 

exchange. 

When simulating MQ-CEST data it is important to take B1 field inhomogeneity into 

account. To do this, for each CEST offset frequency, the CEST intensity is calculated for ten 

B1 fields evenly spaced between ±2σ of the average CEST pulse strength15 (in simulations σ is 

set to 10% of the CEST pulse strength while for fitting experimentally determined values are 

used) and the final intensity result is calculated as the sum of the results weighted by Gaussian 

coefficients.  

For fitting experimental data, the results of Eq. 5 need to be passed to an optimiser to 

minimise the standard χ2 equation16: 

 

𝜒(𝑥)3 = ∑
45()*,,6	5-./-,,(9);

"

<()*,,
"

=
>?!          (6) 

 

Here the sum extends over all experimental data points per residue, including data at multiple 

CEST field strengths and/or multiple static magnetic field strengths, and x refers to the 

parameters to be optimised, including kex, Δω, average 15Nη chemical shifts and relaxation rates. 

The uncertainties, 𝜎@AB,> are estimated from the scatter in the CEST profiles17. A python script, 

which makes extensive use of the numpy and scipy modules13 has been written for the 

simulation and fitting of MQ-CEST data. The least-squares optimiser is implemented using the 

LMFIT module18. This script is available upon request. 

 

Analysis of longitudinal zz-exchange spectra. For each temperature the intensities of the 

cross and diagonal peaks for the Nη nuclei are quantified using FuDA19. They are then fit to the 

following equation with kex as a fitting parameter1: 

 
5.0D50.
5..D500

=	 !6	@AB	(6E()()
!D@AB	(6E()()

	         (7) 

 

Where Iab and Iba are the cross-peak intensities and Iaa and Ibb are the intensities of the two 

diagonal peaks.  

Supporting Table 
 

Table S1: Acquisition parameters used in MQ-CEST NMR Experiments (all experiments 
were carried out using the 13C detect pulse sequence in Fig 1D) 

a) TXI: HCN inverse room-temperature probe. TCI: Cryogenic HNC inverse probe with cooled 1H and 
13C preamplifiers, 1H inner coil and 13C outer coil. TXO: 13C-optimised cryogenic CNH probe with 
cooled 1H, 13C and 15N preamplifiers, 13C inner coil and 1H outer coil. 
b) Shown are values calibrated as described above. 
c) Recovery delay. 
d) Number of transients (scans) per FID. 
e) Total acquisition time of the experiment. 
  

Temp 
(K) 

B0 Field 
(T), probea 

B1 
Fieldb 

(Hz) 

CEST offsets 15N SW (ppm), 
t1 complex 
points 

d1c 
(sec)  

NSd Timee 
(hours) 

 
50 mM [13C6,15N4] arginine in 50% H2O/50% MeOH. MeOD insert provides lock signal 
258.1 14.1, TXI 10.0  

 
 

24 evenly spaced offsets 
between 66.9-74.1 ppm 

(20 Hz gap) 

 
 
 

9.7 ppm, 
14 complex 

points 

 
 
 
 

1.5 

 
 
 
 

4 

 
 
 
 

1.5 

258.1 14.1, TXI 20.0 
264.0 14.1, TXI 10.0 
264.0 14.1, TXI 20.0 
269.6 14.1, TXI 10.0 
269.6 14.1, TXI 20.0 
275.5 14.1, TXI 10.0 
275.5 14.1, TXI 14.1 
 
1.1 mM [U-13C,15N]-labelled L99A mutant of T4 Lysozyme 
293.0 14.1, TXO 10.9 39 evenly spaced offsets 

between 63.9-76.4ppm 
(20 Hz gap) 

8.5 ppm, 
12 complex 

points 

2.0 48 32 

293.0 14.1, TXO 18.9 39 evenly spaced offsets 
between 63.9-76.4ppm 

(20 Hz gap) 

8.5 ppm, 
12 complex 

points 

2.0 40 26.5 

293.0 16.4, TCI 10.6 42 evenly spaced offsets 
between 65-76.5 ppm 

(20 Hz gap) 

8.5 ppm, 14 
complex points 

2.0 72 58.5 

293.0 18.8, TCI 10.7 46 evenly spaced offsets 
between 65.5-76.6 ppm 

(20 Hz gap) 

8.5 ppm, 16 
complex points 

2.0 80 83 

293.0 22.3, TCI 11.2 55 evenly spaced offsets 
between 65.3-76.5 ppm 

(20 Hz gap) 

8.7 ppm, 19 
complex points 

2.0 40 58 

Table S2: Parameters obtained from MQ-CEST for L99A T4 Lysozyme at 293 K 
Residue kex (s-1)a Δω (ppm)a 
R8 814±65 1.54±0.04 
R14 1722±61 2.76±0.04 
R52 19±2 1.19±0.01 
R76 892±18 1.92±0.01 
R80 1184±75 1.43±0.04 
R95 <2.7 5.41±0.02 
R96 156±4 1.98±0.01 
R119 788±26 1.61±0.02 
R125 558±32 1.79±0.02 
R137 1106±21 1.98±0.02 
R154 518±14 1.64±0.01 

 
a) In all cases errors are taken from the covariance matrix of the fits16. These errors are cross validated 
by performing fits using grid searched kex values (see Fig S8). 

Supporting Figures 
 

 
Fig S1. Simulation of MQ-CEST profiles with varying Δω and kex. In both cases the static magnetic 
field strength (B0) is 16.4 T, the CEST B1 strength is 10 Hz (inhomogeneity of 1 Hz) and is applied for 
TCEST = 250 ms. The average of the two 15Nη chemical shifts is set to 71 ppm. (A) The exchange rate is 
constant, kex = 100 s-1, and Δω is varied (B) The chemical shift difference is kept constant, Δω = 2 ppm, 
and the exchange rate kex is varied.   

Fig S2. Correlation plot of fitted kex versus kex used as input for the simulated data under different 
conditions. In each of the 12 plots, each datapoint represents the average (circle) and standard deviation 
(vertical line) of the kex values obtained from 100 simulated datasets under different conditions (kex, Δω, 
B0 and B1). Random Gaussian noise with standard deviation equal to 2% of the signal intensity at each 
point was added to each simulation prior to the fit. (A) A B0 field strength of 16.4 T and a B1 strength 
of 10 Hz (red) were used to simulate the data. (B) A B0 field strength of 16.4 T and B1 field strengths of 
10, 15 and 20 Hz were used (green). (C) B0 field strengths of 14.1 T, 16.4 T, 18.8 T, and 22.3 T and a 
B1 field strength of 10 Hz were used. For all simulations a saturation pulse length of TCEST = 250 ms was 
used and assuming a B1 field inhomogeneity of 10%. For each calculation 10 B1 fields are evenly space 
between the mean B1 field strength ±2σ and weighted according to a Gaussian distribution.  
  



S10 
 

 
 
Fig S3. Correlation plot of fitted Δω versus Δω used as input for the simulated data under different 
conditions. In each of the 12 plots, each datapoint represents the average (circle) and standard deviation 
(vertical line) of the Δω values obtained from 100 simulated datasets under different conditions (kex, 
Δω, B0 and B1). Random Gaussian noise with standard deviation equal to 2% of the signal intensity at 
each point was added to each simulation prior to the fit. (A) A B0 field strength of 16.4 T and a B1 
strength of 10 Hz (red) were used to simulate the data. (B) A B0 field strength of 16.4 T and B1 field 
strengths of 10, 15 and 20 Hz were used (green). (C) B0 field strengths of 14.1 T, 16.4 T, 18.8 T, and 
22.3 T and a B1 field strength of 10 Hz were used. For all simulations a saturation pulse length of TCEST 

= 250 ms was used and assuming a B1 field inhomogeneity of 10%. For each calculation 10 B1 fields 
are evenly space between the mean B1 field strength ±2σ and weighted according to a Gaussian 
distribution.  

Fig S4. Pulse sequences for performing the MQ-CEST experiment with (A) CEST profiles encoded 
onto 13Cζ-15NƐ coherences, (B) CEST profiles encoded onto 1Hε-15Nε resonances and (C) CEST profiles 
encoded onto 13Cζ-15Nη double quantum coherences2. The carrier frequencies are set to 157 ppm (13C) 
in all cases, (A) and (B) 84 ppm (15N), (C) 71 ppm (15N), and (A) and (C) 7.15 ppm (1H), (B) 4.77 ppm 
(1H) except for during the CEST period when it is moved to 8 ppm (1H) for this sequence. Narrow and 
wide bars denote 90° and 180° pulses, respectively with all pulses applied along x unless otherwise 
indicated. The bell shapes denote shaped and frequency-selective pulses with the letter indicating the 
type: S refers to a Seduce pulse20, E to an EBURP pulse21, R to a REBURP pulse and W to a sinc pulse. 
Shaped pulses are applied to the nuclei indicated above the pulse and all sinc pulses are applied to the 
water resonance. The shaped pulses are applied for durations of 300 μs (S), 1.5 ms (E), 4.5 ms (R) and 
1 ms (W). Striped boxes indicate decoupling or CEST periods with the text indicating the scheme used: 
CPD (composite decoupling scheme of 90x240y90x applied at 6.25 kHz), CW (continuous wave, power 
set as desired for the CEST experiment typically 10-20 Hz and applied at CEST offset frequency for 
250 ms), WALTZ22 (3.125 KHz), GARP23 (0.7 KHz).  The delays are set to: d1 = 1.5 s, τa = 1/(4JHN) = 
2.7 ms and τb = 1/(4JNC) = 12.5 ms. The phase cycle used is ϕ1 = x, -x; ϕ2 = 2(x), 2(-x); and ϕrec = x, -x, 
-x, x. Quadrature in t1 is achieved by States-TPPI of ϕ2

24
.  All gradients are 1 ms in length and applied 

with the following strengths: g1 = 12.3 G/cm, g2 = 5.9 G/cm, g3 = 16.6 G/cm, g4 = 10.2 G/cm, g5 = 
7.0 G/cm, g6 = 15.5 G/cm, g7 = 9.1 G/cm and g8 = 19.8 G/cm. 

Fig S5. Comparison of signal/noise ratios for 13C (Fig S4A) and 1H-detected (Fig S4B) reference planes 
of the MQ-CEST experiment. The signal/noise are calculated for R52 in L99A T4 Lysozyme. The 
comparison here ignores the role of hydrogen-exchange with the solvent (very slow under the conditions 
employed here for R5225). The spectra were acquired with the same number of scans, complex points 
(and acquisition time) in t1 and recycle delay on a 700 MHz spectrometer equipped with a TCI cryogenic 
probe and processed identically. The temperature of the sample was altered in steps of 5 K between 283 
K and 298 K to change the correlation time. The global correlation time, 𝜏!, is estimated at each 
temperature using an empirical relationship26. The local correlation time is then estimated from this as 
𝑆"𝜏! with the value for the 15N-1H order parameters S2 taken from Werbeck et al27. The measured signal 
to noise ratios were plotted against the estimated local correlation times above. The dotted lines indicate 
fits to a decaying exponential function. This fit is extrapolated between 1 and 25 ns. Overall, 13C 
detection is preferable when the local correlation time of a residue exceeds approximately 10 ns. On a 
13C optimised TXO cryogenic probe, 13C detection is preferable for residues with even smaller local 
correlation times. Of note is that flexible arginine side chains, with small local correlation times, are 
often severely affected by hydrogen-exchange.  

Fig S6. MQ-CEST data for L99A T4 Lysozyme collected at 293 K. In all cases the CEST field is applied 
for TCEST = 250 ms with a field strength of approximately 10 Hz (the exact B1 field is calibrated in all 
cases using the method of Guenneugues4). The data is additionally fitted with an experiment recorded 
with a CEST field of 18 Hz at 14.1 T (data not shown for clarity).  

Fig S7. Calculated transverse relaxation rate of 15Nη as a function of the static magnetic field strength 
(B0) and correlation time (τC). The calculated relaxation rates are the sum of dipolar relaxation with the 
two 1Hη nuclei12 and relaxation due to chemical shift anisotropy (CSA) of the 15Nη nuclei (the CSA is 
set to 75 ppm, which is estimated from DFT calculations). Dipolar relaxation between 15Nη and other 
nuclei is not included and is confirmed to have a minimal impact on the results. Overall, the calculated 
relaxation rates show very little dependence on the B0 field strength.  

Fig S8. Contour plot showing 𝜒" values as a function of R2 and kex. The colors indicate different 
confidence intervals calculated by taking the difference between the minimum 𝜒" value and the 𝜒" 
value at each point in the grid and then drawing contour lines for Δ𝜒" at the indicated confidence 
intervals for two degrees of freedom16. Overall, the fitted exchange rate, kex, is only very weakly 
correlated with the transverse relaxation rates, R2. This means that accurate kex are obtained although 
the obtained R2 values may be uncertain. The horizontal dashed lines represent the confidence interval 
for the experimentally measured R2 values ( R2 ± σR2 ). 

Fig S9. Comparison of fitted kex rates when R2 is fixed to the experimentally determined value (x-axis) 
and when it is allowed float in the fitting process (y-axis). The orange dashed line is x = y.  

Pulse Sequences 

; Filename: arg_hczne_cest.gk 
; 
; (He)CzNe HSQC with CEST 
; 1H excitation, 13C detection 
; semi-constant time evolution in t1 
; Gogulan Karunanithy 05/11/19 
; 
; This pulse sequence is for recording MQ CEST profiles of the guanidinium groups of arginine residues 
; in uniformly labelled {13C-15N} samples. In this experiment magnetisation is excited on 1Hepsilon and  
; detected on 13Czeta.  
; 
; WARNING: This sequence does not work with ‘getprosol’. If you type ‘getprosol’ by accident when using  
; this sequence all parameters must be checked carefully manually.  
; 
; start on He->Ne(semi-constant time evolution)-> 
;          2CzetaNe -> 4CzetaNhNh (CEST) -> 
;          C(x) for detection 
; 
; How to: 
; -------- 
; --------  
; Set 13C carrier frequency to 157 ppm (C zeta)  
; Set 1H carrier frequency to 7.15 ppm (Heps) 
; Set 15N carrier frequency to 84 ppm (Neps) 
; 
; Initial 1Heps-15Neps INEPT: 
;  - Set cnst2 = 92 (Heps-Neps J-coupling) 
; 
; Semi Constant Time Evolution: 
; - Set cnst3 = 20 (Neps-Czeta J-coupling) 
; - Set cnst12 = 40.5 (Cdelta chemical shift) 
; - Apply 300 us Seduce inversion pulse at 16.4 T on Cdelta (set by cnst12) and Czeta (on resonance) to  
;   refocus and evolve respective J-couplings 
; - Apply 80 us WALTZ64 decoupling on 1H to suppress J-coupling revolution 
; - At the end of this period we will have 2Cz(zeta)Nz(eps) magnetisation – note that sweep widths may need 
;   to be adjusted slightly to avoid negative delays.  
;  
; Selective 13Czeta Excitation 
; - 1.5 ms Eburp2 pulse at 16.4 T 
; 
; CEST period 
; - The saturation frequencies are taken from fq1list. If the absolute frequency value is set to >10000Hz a  
;   references plane is recorded 
; - The CEST pulse length is set by d17. The strength of the CEST pulse (pl37) must be carefully  
;   calibrated: we use the method of Guenneugues et al. Journal of Magnetic Resonance, 136, 118-126, 1999 
; - 40 us Composite decoupling (90x240y90x) is applied to 1H during CEST 
; 
; During final INEPT a 4.5 ms ReBurp inversion is applied to Neta at 16.4 T. cnst32 should be set to 70 ppm  
; (position of Neta protons) 
; 
; 80 us WALTZ64 decoupling and 350 us GARP4 deoupling are applied to 1H and 15N respectively during  
; acquisition  
 
#include <Avance.incl> 
#include <Delay.incl> 
#include <Grad.incl> 
 
;DEFINE PULSES 
 
define pulse pwc 
        "pwc=p1"                                  ;13C hard pulse at pl1 
define pulse pwc_selr1 
        "pwc_selr1=p11"                           ;13C selective seduce-pulse at pl11 (acts on C zeta) 
define pulse pwc_selr2 
        "pwc_selr2=p12"                           ;13C selective seduce-pulse at pl12 (acts on C delta) 
define pulse pwc_sele 
        "pwc_sele=p13"                            ;13C selective eburp-pulse at pl13 
 
define pulse pwh 
        "pwh=p2"                                  ;1H hard pulse at pl2 
 
define pulse pwn 
        "pwn=p3"                                  ;15N hard pulse at pl3 
define pulse pwn_selr 
        "pwn_selr=p31"                            ;15N selective r-pulse at pl31 
 
;DEFINE DELAYS 
 
define delay taua 
        "taua=1s/(cnst2*4)"                       ;1/4JHN 
define delay taub 
        "taub=1s/(cnst3*4)"                       ;1/4JCN 

"d11= 30m"                                        ;Delay for disk 
"d12= 4u"                                         ;Delay for power switching 
"d16= 200u"                                       ;Delay for gradient recovery 
 
"in0=inf1/2"                                      ;incremental delay 
 
#ifdef HALFDWELL 
    "d0=in0/2" 
#else 
    "d0=0.0" 
#endif /*HALFDWELL*/ 
 
;DEFINE OFFSETS 
 
define list<frequency> N15sat = <$FQ1LIST> 
 
"spoffs11=0"                                      ;Czeta seduce 
"spoffs12=(cnst12*bf1)/1000000-o1"                ;Cdelta  seduce 
"spoffs13=0"                                      ;Czeta eburp2 
"spoffs31=0"                                      ;15N reburp 
 
;DEFINE ZERO POWER ON ALL CHANNELS USING SHAPED PULSES 
 
"plw10=0" 
"plw20=0" 
"plw30=0" 
 
;DEFINE CONSTANTS 
 
"cnst11=o1/bf1" 
"cnst21=o2/bf2" 
 
"cnst31=o3/bf3"                                   ; Ne 
"cnst33=0.5*(cnst31+cnst32)"                      ;15N decoupling position 
 
;PULSE PROGRAM BEGINS 
 
1 ze 
2 d11 do:f2 do:f3             ;decoupling off 
  if "abs(N15sat)>10000Hz" { 
    d12 pl37:f3 pl22:f2       ; set appropriate power level for 15N CEST pulse and hydrogen decoupling 
    30u fq=N15sat:f3          ; put frequency at right place  
 
    30u fq=0:f2               ; 
    d12 cpd8:f3 cpd5:f2       ; turn on CEST (N15) and decoupling (1H) 
    d17                       ; delay for CEST 
    d12 do:f2 do:f3 
  } 
 
  50u UNBLKGRAD 
  d12 pl2:f2 
  (pwh ph10):f2 
 
  2u 
  p51:gp1                                         ;cleaning gradient 
  d16 
 
  50u BLKGRAD 
 
  d12 pl29:f2 
 
  d1  
  
  ;start purge equilibrium 15N magnetisation 
 
  50u UNBLKGRAD                                   ;gradient amp on 
  d12 fq=cnst33(bf ppm):f3                        ;set 15N carrier 
  d12 pl2:f2 pl3:f3                               ;power to high (1H, 15N) 
  (pwn ph10):f3                                   ;90x on Nitrogen to purge 
  2u 
  p51:gp1                                         ;cleaning gradient 
  d16 
 
  d12 fq=cnst31(bf ppm):f3                        ;set 15N carrier, Nepsilon 
  d12 fq=cnst11(bf ppm):f1 
  d12 fq=cnst21(bf ppm):f2 
 
  ;end purge block 
 
  ;start HN INEPT (Non-selective) 
 
  (pwh ph1):f2 
  "DELTA = taua-0.6366*pwh-2u-p52-d16-0.5*larger(pwh*2,pwn*2)" 
  DELTA 
  2u 
  p52:gp2 
  d16 

  ( center (pwh*2 ph10):f2 (pwn*2 ph10):f3 ) 
  2u 
  p52:gp2 
  d16 
  "DELTA = taua-0.5*larger(pwc*2,pwn*2)-2u-p52-d16-0.6366*pwh" 
  DELTA 
  (pwh ph11):f2 
  ; end INEPT now on HzNz 
 
  2u 
  p53:gp3                                         ;cleaning gradient 
  d16 
  d12 pl10:f1 pl29:f2                          ;power to 0(13C), decouple (1H): prepare for what's coming 
 
 
 ;now go from 2HzNz to 2CzNz with semi-constant time evolution 
 (pwn ph2):f3       ;make 2HzNy (phase cycle for 2D) 
  
 if "d0<(2*taua - pwc_selr2)" 
  { 
  "DELTA = d0" 
  DELTA 
  (pwc_selr2:sp12 ph10):f1        ; 180 on Cdelta  
  "DELTA = 2*taua - d0 - pwc_selr2" 
  DELTA  
  d12 cpd2:f2  
  "DELTA = taub + d0 - pwc_selr1 - d12 - 2*taua" 
  DELTA 
  (pwc_selr1:sp11 ph10):f1        ; 180 on Czeta 
  }  
 else 
  { 
  "DELTA = 2*taua" 
  DELTA 
  d12 cpd2:f2 
  "DELTA = d0 - 2*taua" 
  DELTA 
  (pwc_selr2:sp12 ph10):f1        ; 180 on Cdelta   
  "DELTA = taub - pwc_selr2 - pwc_selr1 - d12" 
  DELTA 
  (pwc_selr1:sp11 ph10):f1        ; 180 on Czeta 
  }  
 
if "d0<taub" 
 { 
 (pwn*2.0 ph10):f3 
 "DELTA = taub - d0" 
 DELTA 
 } 
else 
 { 
 "DELTA = d0 - taub" 
 DELTA 
 (pwn*2.0 ph10):f3 
 } 
 
 
 (pwn ph10):f3     ; go back to 2CzNz 
  d12 do:f2     ; turn off proton decoupling 
 
  2u 
  p54:gp4                                         ;cleaning gradient 
  d16 
  30u fq=cnst33(bf ppm):f3 
  d12 pl10:f1 
  d12 
  (pwc_sele:sp13 ph10):f1                          ;13C selective excitation (eburp pulse) 
                                                  ;start Nh selective block 
  "DELTA = taub-2u-p52-d16-0.5*larger(pwc*2,2*pwn)" 
  DELTA 
  2u 
  p55:gp5                                         ;p55 gradient time 
  d16 pl1:f1                                      ;power to high (13C) d16 is 200 us delay for grad 
recovery 
  ( center (pwc*2 ph10):f1 (pwn*2 ph10):f3 ) ; 15N hard  
  2u 
  p55:gp5                                         ;p53: gradient time 
  d16 
  "DELTA = taub-2u-p52-d16-0.5*larger(pwc*2,2*pwn)-0.6366*pwc" 
  DELTA 
                                                  ;end INEPT 
  (pwc ph11):f1                                   ;We should now be on 4CzNhzNhz ( 
 
  ; do CEST here 
 
  if "abs(N15sat)>10000Hz" { 
    d12 pl30:f3              ; for reference plane 
  } else { 

    d12 pl37:f3 pl22:f2    ; set appropriate power level for 15N CEST pulse and hydrogen decoupling 
  } 
  30u fq=N15sat:f3          ; stick frequency at right place  
  d12 cpd8:f3 cpd5:f2         ; turn on CEST (N15) and decoupling (1H)  
 
  if "abs(N15sat)>10000Hz" { 
    d12                     ; set delay v short for reference plane 
}  
else { 
  d17                       ; delay for CEST 
  }   
  d12 do:f3 do:f2           ; turn off decoupler and cest pulse 
  30u fq=cnst32(bf ppm):f3 ; place carrier back in right spot for N15 i.e. Nh 
 
  ; end CEST  
 
  2u 
  p56:gp6                                         ;cleaning gradient 
  d16 
 
  d12 pl30:f3 pl29:f2                                      ;power for 1H decoupler and 15N shape 
  d12 
 
 
 (pwc ph10):f1                                   ;-4CzNh1zNh2z to 4CyNh1zNh2z 
                                                   
  "DELTA = taub-0.6366*pwc-2u-p55-d16-0.5*larger(pwc*2,pwn_selr)" 
  DELTA 
  2u 
  p57:gp7 
  d16 
  ( center (pwc*2 ph10):f1 (pwn_selr:sp31 ph10):f3 ) 
  2u 
  p57:gp7 
  d16 
  30u fq=cnst33(bf ppm):f3                        ;set 15N carrier to decouple (in middle) 
  d12 pl39:f3                                     ;power to decouple(15N) 
  50u BLKGRAD                                     ;gradient amp off 
  "DELTA = taub-0.5*larger(2*pwc,pwn_selr)-2u-p55-d16-30u-d12-50u" 
  DELTA 
                                                  ;end Nh selective block 
 
  ;end second INEPT, acquire Cy magnetisation with 15N, 1H decoupling 
 
  go=2 ph31 cpd2:f2 cpd3:f3 
    d11 do:f2 do:f3 mc #0 to 2 
    F2QF(calclist(N15sat,1)) 
    F1PH(calph(ph2,+90), caldel(d0,+in0)) 
 
exit 
 
;PHASE PROGRAMS 
 
ph1=  0 2 
ph2=  0 0 2 2 
ph31= 0 2 2 0 
 
ph10= 0 
ph11= 1 
ph12= 2 
ph13= 3 
 
;DEFINITIONS 
 
;pl1     : f1 channel - power level for hard pulse 
;pl2     : f2 channel - power level for hard pulse 
;pl3     : f3 channel - power level for hard pulse 
;pl22    : f2 channel - power level for composite 1H decoupling 
;pl29    : f2 channel - power level for CPD/BB decoupling 
;pl37    : f3 channel - power level for CEST pulse 
;pl39    : f3 channel - power level for CPD/BB decoupling 
;sp11    : f1 channel - shaped pulse  180 degree (selective for Czeta) 
;spnam11 : Seduce.100 
;sp12    : f1 channel - shaped pulse  180 degree (selective for Cdelta) 
;spnam12 : Seduce.100 
;sp13    : f1 channel - shaped pulse  90 degree (selective for Czeta) 
;spnam13 : Eburp2.1000 
;sp31    : f3 channel - shaped pulse 180 degree (selective for Ne) 
;spnam31 : Reburp.1000 
;p1      : f1 channel - 90 degree high power pulse 
;p10     : setup for composite decoupling                      [40 us] 
;p11     : f1 channel - 180 degree shaped pulse                [300 us at 18.8T] 
;p12     : f1 channel - 180 degree shaped pulse                [300 us at 18.8T] 
;p13     : f1 channel - 90 degree shaped pulse                 [1.5 ms at 18.8T] 
;p2      : f2 channel - 90 degree high power pulse 
;p25     : f2 channel - composite decoupling 90 degree pulse  [40 us] 
;p3      : f3 channel - 90 degree high power pulse 
;p31     : f3 channel - 180 degree shaped pulse               [4.5 ms at 18.8T] 

;p51     : homospoil pulse 
;p52     : gradient pulse 
;p53     : homospoil pulse 
;p54     : homospoil pulse 
;p55     : gradient pulse 
;p56     : homospoil pulse 
;d0      : incremented delay 
;d1      : relaxation delay 
;taua    : 1/(4JHN) 
;taub    : 1/(4JCN) 
;d11     : delay for disk I/O                                            [30 ms] 
;d12     : delay for power switching                                      [2 us] 
;d16     : delay for homospoil/gradient recovery                        [200 us] 
;d17     : delay for CEST                                               [250 ms] 
;cnst2   : J(HN)                                                         [92 Hz] 
;cnst3   : J(CN)                                                         [20 Hz] 
;cnst12  : Cdelta chemical shift offset                               [40.5 ppm] 
;cnst31  : Ne chemical shift offset                      [= o3p, 84.0 ppm] 
;cnst32  : Nh chemical shift offset                                   [70 ppm] 
;cnst33  : average of Nh and Ne shifts                         [~77 ppm] 
;o1p     : Czeta chemical shift offset                                 [156 ppm] 
;o2p     : He/Hh chemical shift offset                           [7 ppm] 
;o3p     : Ne chemical shift offset                               [84 ppm] 
;inf2    : 1/SW = 2 * DW 
;in0     : 1/(2 * SW) = DW 
;NS      : 8 * n 
;DS      : 64 
;td1     : number of experiments 
;FnMODE  : States-TPPI 
;cpd2    : decoupling according to sequence defined by cpdprg2 (WALTZ64) 
;pcpd2   : f2 channel - 90 degree pulse for decoupling sequence          [80 us] 
;cpd3    : decoupling according to sequence defined by cpdprg3 (GARP4) 
;pcpd3   : f3 channel - 90 degree pulse for decoupling sequence         [350 us] 
;cpd5    : decoupling according to sequence defined by cpdprg5 (composite) 
;cpd8    : sequence for CEST (cwp) 
 
;for z-only gradients: 
;gpz1    : 23% (spoil) 
;gpz2    : 11% 
;gpz3    : 31% (spoil) 
;gpz4    : 19% (spoil) 
;gpz5    : 13%  
;gpz6    : 29% (spoil) 
;gpz7    : 17% 
 
;use gradient files: 
;gpnam1  : SMSQ10.100 
;gpnam2  : SMSQ10.100 
;gpnam3  : SMSQ10.100 
;gpnam4  : SMSQ10.100 
;gpnam5  : SMSQ10.100 
;gpnam6  : SMSQ10.100 
;gpnam7  : SMSQ10.100 
 
;preprocessor-flags-start 
;HALFDWELL: for initial sampling delay of half a dwell-time with 
;           option -DHALFDWELL (eda: ZGOPTNS) 
;preprocessor-flags-end 

; Filename: arg_hnhcesth_cest.gk 
; 
; H-Ne HSQC with CEST on 4CzZNhZNhZ 
; 1H excitation and detection 
; semi-constant time evolution in t1 
; Gogulan Karunanithy 20/11/19 
; 
; start on He->Ne(semi-constant time evolution)-> 
;          2CzNe -> 4CzNhNh (CEST) -> 2CzNe -> Ne -> He Detection 
; 
; How to: 
; -------- 
; --------  
; Set 1H carrier frequency to 4.77 ppm (Water) 
; Set 13C carrier frequency to 157 ppm (C zeta) 
; Set 15N carrier frequency to 84 ppm (Neps) 
;  
; 
; Initial 1Heps-15Neps INEPT: 
;  - Set cnst2 = 92 (Heps-Neps J-coupling) 
; 
; Semi Constant Time Evolution: 
; - Set cnst3 = 20 (Neps-Czeta J-coupling) 
; - Set cnst22 = 40.5 (Cdelta chemical shift) 
; - Apply 300 us Seduce inversion pulse at 16.4 T on Cdelta (set by cnst12) and Czeta (on resonance) to  
;   refocus and evolve respective J-couplings 
; - Apply 80 us WALTZ64 decoupling on 1H to suppress J-coupling revolution 
; - At the end of this period we will have 2Cz(zeta)Nz(eps) magnetisation – note that sweep widths may need 
;   to be adjusted slightly to avoid negative delays.  
;  
; Selective 13Czeta Excitation 
; - 1.5 ms Eburp2 pulse at 16.4 T 
; 
; CEST period 
; - The saturation frequencies are taken from fq1list. If the absolute frequency value is set to >10000Hz a  
;   references plane is recorded 
; - Set cnst12 = 8.0 (move 1H carrier to middle of amide region) 
; - The CEST pulse length is set by d17. The strength of the CEST pulse (pl37) must be carefully  
;   calibrated: we use the method of Guenneugues et al. Journal of Magnetic Resonance, 136, 118-126, 1999 
; - 40 us Composite decoupling (90x240y90x) is applied to 1H during CEST 
; 
; 1000 us Sinc pulses are employed for water suppression at 16.4 T 
; 
; 350 us GARP4 decoupling is applied 15N respectively during acquisition  
 
#include <Avance.incl> 
#include <Delay.incl> 
#include <Grad.incl> 
 
;DEFINE PULSES 
 
define pulse pwc 
        "pwc=p2"                                  ;13C hard pulse at pl1 
define pulse pwc_selr1 
        "pwc_selr1=p21"                           ;13C selective seduce-pulse at pl11 (acts on C zeta) 
define pulse pwc_selr2 
        "pwc_selr2=p22"                           ;13C selective seduce-pulse at pl12 (acts on C delta) 
define pulse pwc_sele 
        "pwc_sele=p23"                            ;13C selective eburp-pulse at pl13 
 
define pulse pwh 
        "pwh=p1"                                  ;1H hard pulse at pl2 
 
define pulse pwn 
        "pwn=p3"                                  ;15N hard pulse at pl3 
define pulse pwn_selr 
        "pwn_selr=p31"                            ;15N selective r-pulse at pl31 
 
;DEFINE DELAYS 
 
define delay taua 
        "taua=1s/(cnst2*4)"                       ;1/4JHN 
define delay taub 
        "taub=1s/(cnst3*4)"                       ;1/4JCN 
 
"d11= 30m"                                        ;Delay for disk 
"d12= 4u"                                         ;Delay for power switching 
"d16= 200u"                                       ;Delay for gradient recovery 
 
"in0=inf1/2"                                      ;incremental delay 
 
#ifdef HALFDWELL 
    "d0=in0/2" 
#else 
    "d0=0.0" 
#endif /*HALFDWELL*/ 
 
;DEFINE ZERO POWER ON ALL CHANNELS USING SHAPED PULSES 

"plw10=0" 
"plw20=0" 
"plw30=0" 
 
;DEFINE CONSTANTS 
"cnst11=o1/bf1" 
"cnst21=o2/bf2" 
"cnst31=o3/bf3"                                   ;Nepsilon 
"cnst33=0.5*(cnst31+cnst32)"                      ;15N decoupling position 
 
;DEFINE OFFSETS 
 
define list<frequency> N15sat = <$FQ1LIST> 
 
"spoffs21=0"                                      ;Czeta seduce 
"spoffs22=(cnst22*bf2)/1000000-o2"       ;Cdelta seduce 
 
;PULSE PROGRAM BEGINS 
 
1 ze 
2 d11 do:f3                                 ;decoupling off 
 
 if "abs(N15sat)>10000Hz" { 
    d12 pl37:f3 pl12:f1    ; set appropriate power level for 15N CEST pulse and hydrogen decoupling 
    30u fq=N15sat:f3       ; put 15N frequency at right place 
 
   4u fq=cnst12(bf ppm):f1            
   (p15 ph11):f1 
   d12 cpd8:f3 cpds5:f1 ph10      ; turn on CEST (N15) and decoupling (1H)  
   d17                       ; delay for CEST 
   d12 do:f1 do:f3 
   (p15 ph13):f1   
  } 
 
  50u UNBLKGRAD 
  d12 pl1:f1 ; 
  (pwh ph10):f1     ; purge any 1H magnetisation prior to d1: always start from same place 
 
  2u 
  p51:gp1                                         ;cleaning gradient 
  d16 
 
  (pwh ph11):f1     ; purge any 1H magnetisation prior to d1: always start from same place 
 
  2u 
  p51:gp8                                         ;cleaning gradient 
  d16 
 
  50u BLKGRAD 
 
  d1 
 
  ;start purge equilibrium 15N magnetisation 
 
  50u UNBLKGRAD                                   ;gradient amp on 
 
  d12 fq=cnst33(bf ppm):f3                        ;set 15N carrier 
  d12 fq=0:f1                        ;set 1H onto water 
 
  d12 pl1:f1 pl3:f3                               ;power to high (1H, 15N) 
 (pwn ph10):f3                                   ;90x on Nitrogen to purge 
 
  2u 
  p51:gp1                                         ;cleaning gradient 
  d16 
 
  d12 fq=cnst21(bf ppm):f2                        ;set to Czeta 
 
  d12 fq=0:f2 
  d12 fq=0:f3 
 
  ;end purge block 
 
  ;start HN INEPT (Non-selective) 
 
 
  d12 pl1:f1 
 
  (pwh ph10):f1 
 
  "DELTA = taua-0.6366*pwh-2u-p52-d16-larger(pwh,pwn)" 
  DELTA 
 
  2u 
  p52:gp2 
  d16 

  ( center (pwh*2 ph10):f1 (pwn*2 ph10):f3 ) 
 
  2u 
  p52:gp2 
  d16 
 
  "DELTA = taua-larger(pwn,pwh)-2u-p52-d16-0.6366*pwh" 
  DELTA 
 
  (pwh ph11):f1 
  d12 pl10:f1 
  (p11:sp11 ph10):f1      ;water suppression pulse 
  2u 
 
  ; end INEPT now on HzNz 
 
  2u 
  p53:gp3                                         ;cleaning gradient 
  d16 
 
  d12 pl20:f2 pl19:f1                            ;power to 0(13C) decoup on 1H 
 
 ;now go from 2HzNz to 2CzNz with semi-constant time evolution 
 (pwn ph2):f3       ;make 2HzNy (phase cycle for 2D) 
 
 
 if "d0<(2*taua - pwc_selr2)" 
  { 
  "DELTA = d0" 
  DELTA 
  (pwc_selr2:sp22 ph10):f2        ; 180 on Cdelta  
  "DELTA = 2*taua - d0 - pwc_selr2" 
  DELTA  
  (p19 ph11):f1 
  d12 cpds1:f1 ph10 
  "DELTA = taub + d0 - pwc_selr1 - p19 - d12 - 2*taua" 
  DELTA 
  (pwc_selr1:sp21 ph10):f2        ; 180 on Czeta 
  }  
 else 
  { 
  "DELTA = 2*taua" 
  DELTA 
  (p19 ph11):f1 
  d12 cpds1:f1 ph10 
  "DELTA = d0 - 2*taua" 
  DELTA 
  (pwc_selr2:sp22 ph10):f2        ; 180 on Cdelta   
  "DELTA = taub - pwc_selr2 - pwc_selr1 - p19 - d12" 
  DELTA 
  (pwc_selr1:sp21 ph10):f2        ; 180 on Czeta 
  }  
 
if "d0<taub" 
 { 
 (pwn*2.0 ph10):f3 
 "DELTA = taub - d0" 
 DELTA 
 } 
else 
 { 
 "DELTA = d0 - taub" 
 DELTA 
 (pwn*2.0 ph10):f3 
 } 
 
  (pwn ph10):f3 
 
  d12 do:f1 
  (p19 ph13):f1 
 
  2u 
  p54:gp4                                         ;cleaning gradient 
  d16 
  
(pwc_sele:sp23 ph1):f2                          ;13C selective excitation (eburp pulse) 
 
 
 "DELTA = taub-2u-p55-d16-larger(pwc,pwn)" 
  DELTA 
 
  2u 
  p55:gp5                                         ;p55 gradient time 
  d16 pl2:f2                                      ;power to high (13C) d16 is 200 us delay for grad 
recovery 
 
 ( center (pwc*2 ph10):f2 (pwn*2 ph10):f3 )      ; 15N hard 

  2u 
  p55:gp5                                         ;p53: gradient time 
  d16 
 
  "DELTA = taub-2u-p55-d16-larger(pwc,pwn)-0.6366*pwc" 
  DELTA 
; 
  (pwc ph11):f2                                   ;We should now be on 4CzNzNz (for N etas) 
 
; CEST here  
  
  if "abs(N15sat)>10000Hz" { 
    d12 pl30:f3  pl10:f1             ; for reference plane 
  } else { 
    d12 pl37:f3 pl12:f1    ; set appropriate power level for 15N CEST pulse and hydrogen decoupling 
  } 
  30u fq=N15sat:f3          ; stick frequency at right place  
  4u fq=cnst12(bf ppm):f1  
  d12 cpd8:f3 cpds5:f1 ph10         ; turn on CEST (N15) and decoupling (1H) - we want high pwr (10kHz) 
composite (90x240y90x) decoupling on proton 
 
  if "abs(N15sat)>10000Hz" { 
    d12                     ; set delay v short for reference plane 
}  
else { 
  d17                       ; delay for CEST 
  }   
  d12 do:f3 do:f1           ; turn off decoupler and cest pulse 
  4u fq=0:f1  
  30u fq=cnst33(bf ppm):f3 ; place carrier back in right spot for N15  
  ; end CEST  
 
  2u 
  p56:gp6                                         ;cleaning gradient 
  d16 
 
  d12 pl3:f3                                      ;reset 15N power 
 
; now go to 2CzNeZ 
  (pwc ph10):f2                                   ;-4CzNh1zNh2z to 4CyNh1zNh2z 
 
  "DELTA = taub-0.6366*pwc-2u-p57-d16-larger(pwc,pwn)" 
  DELTA 
 
  2u 
  p57:gp7 
  d16 
 
  ( center (pwc*2 ph10):f2 (pwn*2 ph10):f3 ) 
 
  2u 
  p57:gp7 
  d16 
 
  "DELTA = taub-larger(pwc,pwn)-2u-p57-d16-0.6366*pwc" 
  DELTA 
 
  (pwc ph11):f2 ; bring Czeta back up: now should be 2CzNz (wrt N epsilon) 
 
 
 2u 
  p58:gp8  
  d16 
 
; go from 2CzNz to pure in phase N epsilon 
 
  d12 pl20:f2 pl19:f1 
  (p19 ph11):f1 
 
  d12 cpds1:f1 ph10 
  (pwn ph10):f3    ; bring down Ne 
 
  "DELTA = taub - 0.5*larger(pwc_selr1,2.0*pwn)" 
  DELTA 
 
  ( center (pwc_selr1:sp21 ph10):f2 (pwn*2 ph10):f3 ) ; refocus coupling to Cdelta and evolve Czeta 
 
  "DELTA = taub - 0.5*larger(pwc_selr1,2.0*pwn) - 2.0*taua - d12" 
  DELTA 
   
  d12 do:f1    ; turn off 1H decoupler  
  (p19 ph13):f1   ; preserve water mag 
 
  "DELTA = 2.0*taua - p19" 
  DELTA 
   
  (pwn ph10):f3    ; bring up HezNez 

  2u 
  p58:gp8 ; purge anything that's not transverse 
  d16 
 
  d12 fq=0:f1                        ;set 1H to water 
 
  d12 pl10:f1 
  (p11:sp11 ph12 ):f1      ;water suppression pulse 
  d12 pl1:f1 
 
 
  (pwh ph10):f1   ; transverse antiphase on proton 
  "DELTA = taua - 0.6366*pwh - 2u - p52- d16 - p11 - d12 - larger(pwh,pwn)" 
  DELTA 
  2u 
  p52:gp2 
  d16 pl10:f1 
  (p11:sp11 ph12):f1      ;water suppression pulse 
  d12 pl1:f1 
  ( center (pwh*2 ph10):f1 (pwn*2 ph10):f3 ) 
  d12 pl10:f1 
  (p11:sp11 ph12):f1      ;water suppression pulse 
  2u 
  p52:gp2 
  d16 pl39:f3                   ;set power for 15N decoupling 
  50u BLKGRAD                                     ;gradient amp off 
  ;end second INEPT, acquire Hy magnetisation with 15N decoupling 
  "DELTA = taua - larger(pwh,pwn)- d12 - p11 - 2u -  p52- d16  - 50u" 
  DELTA 
 
  go=2 ph31 cpd3:f3 
    d11 do:f3 mc #0 to 2 
    F2QF(calclist(N15sat,1)) 
    F1PH(calph(ph2,+90), caldel(d0,+in0)) 
 
exit 
 
;PHASE PROGRAMS 
 
ph1=  0 0 2 2 
ph2=  0 2 
ph31= 0 2 2 0  
 
ph10= 0 
ph11= 1 
ph12= 2 
ph13= 3 
 
;DEFINITIONS 
 
;pl1     : f1 channel - power level for hard pulse 
;pl2     : f2 channel - power level for hard pulse 
;pl3     : f3 channel - power level for hard pulse 
;pl12    : f2 channel - power level for composite 1H decoupling 
;pl19    : f1 channel - power level for CPD/BB decoupling 
;pl37    : f3 channel - power level for CEST pulse 
;pl39    : f3 channel - power level for CPD/BB decoupling 
;sp11    : f1 channel - shaped pulse 90 degree (for water suppression) 
;spnam11 : sinc pulse 
;sp21    : f2 channel - shaped pulse  180 degree (selective for Czeta) 
;spnam21 : Seduce.100 
;sp22    : f2 channel - shaped pulse  180 degree (selective for Cdelta) 
;spnam22 : Seduce.100 
;sp23    : f2 channel - shaped pulse  90 degree (selective for Czeta) 
;spnam23 : Eburp2.1000 
;sp31    : f3 channel - shaped pulse 180 degree (selective for Nepsilon) 
;spnam31 : Reburp.1000 
;p1      : f1 channel - 90 degree high power pulse (1H) 
;p10     : setup for composite decoupling [40us] 
;p11     : f1 channel - water suppression sinc pulse (1000 us) 
;p15     : pulse for 1H CEST decoupling [40us] 
;p19     : pulse for Waltz decoupling [80us] 
;p2      : f2 channel - 90 degree high power pulse (13C) 
;p21     : f2 channel - 180 degree shaped pulse                [300 us at 18.8T] 
;p22     : f2 channel - 180 degree shaped pulse                [300 us at 18.8T] 
;p23     : f2 channel - 90 degree shaped pulse                 [1.5 ms at 18.8T] 
;p3      : f3 channel - 90 degree high power pulse (15N) 
;p31     : f3 channel - 180 degree shaped pulse               [4.5 ms at 18.8T] 
;p51     : homospoil pulse 
;p52     : gradient pulse 
;p53     : homospoil pulse 
;p54     : homospoil pulse 
;p55     : gradient pulse 
;p56     : homospoil pulse 
;p57     : gradient pulse 
;p58     : homospoil pulse 
;p59     : gradient pulse 
;p60     : gradient pulse 

;d0      : incremented delay 
;d1      : relaxation delay 
;taua    : 1/(4JHN) 
;taub    : 1/(4JCN) 
;d11     : delay for disk I/O                                            [30 ms] 
;d12     : delay for power switching                                      [2 us] 
;d16     : delay for homospoil/gradient recovery                        [200 us] 
;d17     : CEST Delay                                                   [250 ms] 
;cnst2   : J(HN)                                                         [92 Hz] 
;cnst3   : J(CN)                                                         [20 Hz] 
;cnst12  : amide chemical shift   [8.0 ppm] 
;cnst22  : Cdelta chemical shift offset                               [40.5 ppm] 
;cnst31  : Nepsilon chemical shift offset                      [= o3p, 84.0 ppm] 
;cnst32  : Neta chemical shift offset                                   [70 ppm] 
;cnst33  : average of Neta and Nepsilon shifts                         [~77 ppm] 
;o1p     : Czeta chemical shift offset                                 [156 ppm] 
;o2p     : He/Hh chemical shift offset                           [7 ppm] 
;o3p     : Ne chemical shift offset                               [84 ppm] 
;inf2    : 1/SW = 2 * DW 
;in0     : 1/(2 * SW) = DW 
;NS      : 4 * n 
;DS      : 64 
;td1     : number of experiments 
;FnMODE  : States-TPPI 
;cpd1    : decoupling according to sequence defined by cpdprg1 (WALTZ64) 
;pcpd1   : f1 channel - 90 degree pulse for decoupling sequence         [80 us] 
;cpd3    : decoupling according to sequence defined by cpdprg3 (GARP4) 
;pcpd3   : f3 channel - 90 degree pulse for decoupling sequence         [350 us] 
;cpd5    : decoupling according to sequence defined by cpdprg5 (composite 90x240y90x) 
;cpd8    : sequence for CEST (cwp) 
 
;for z-only gradients: 
;gpz1    : 41% (spoil) 
;gpz2    : 11% 
;gpz3    : 31% (spoil) 
;gpz4    : 19% (spoil) 
;gpz5    : 13% 
;gpz6    : 29% (spoil) 
;gpz7    : 17% 
;gpz8    : 37% (spoil) 
 
;use gradient files: 
;gpnam1  : SMSQ10.100 
;gpnam2  : SMSQ10.100 
;gpnam3  : SMSQ10.100 
;gpnam4  : SMSQ10.100 
;gpnam5  : SMSQ10.100 
;gpnam6  : SMSQ10.100 
;gpnam7  : SMSQ10.100 
;gpnam8  : SMSQ10.100 
;gpnam9  : SMSQ10.100 
;gpnam10  : SMSQ10.100 
 
;preprocessor-flags-start 
;HALFDWELL: for initial sampling delay of half a dwell-time with 
;           option -DHALFDWELL (eda: ZGOPTNS) 
;preprocessor-flags-end   

